Appendix D

Listing of GAMS Code for WAS

```
*          Water Allocation System (WAS) Model
*                Version 3.2
*
*    This model maximizes the net benefits of water, in Israel, Jordan and
* the Palestinian National Authority (PNA). The model is annual and uses
* the annual renewable amounts of the water sources in the region.
*    This version was jointly made by teams from Israel, Jordan, PNA and
* from Harvard in the Fall of 1997.


* The following line allows for a maximum of 8000 iterations by the solver.
OPTIONS ITERLIM=8000;

* This allows comments to be made in a line by using the characters: {}
$inlinecom{ }

* The maximum number of characters recognized in a single line is 200
$maxcol 200


* The sets or indices used in the model are defined below.

SETS

   i1      all districts with intermediate nodes
   i(i1)   districts without intermediate nodes
   ii(i)   district numbers for Israel
   ip(i)   district numbers for PNA
   ij(i)   district numbers for Jordan
   ig(i)   district number for Golan
   ir(i)   district number for Jerusalem
   in(i1)  node numbers for intermediate nodes
   iii(i1) Israeli districts including intermediate nodes

* The fresh water supplies are described by district in steps
* of increasing production costs, up to five increments

   s       supply source /S1*S5/

* The demands are divided into 3 sectors: Urban, Industrial
* and agricultural

   d       demand sector /URB,IND,AGR/

* Fixed price policies may be described in up to five increments

   steps   fixed price policy steps  /1*5/

* The 2 statements below allow the program to use the indices
* i and j interchangeably, and i1 and j1 interchangeably
```

```
alias (i,j);
alias (i1,j1);
```

**** The following statements add files which include all of the***
**** definitions for the sets declared above.***

**** This file inputs district names and indices for all nodes***

```
$include district.inc
```

**** This file inputs districts without intermediate nodes***

```
$include dstrct_i.inc
```

**** This file inputs districts for Israel***

```
$include dstrctii.inc
```

**** This file inputs districts for PNA***

```
$include dstrct_p.inc
```

**** This file inputs districts for Jordan***

```
$include dstrct_j.inc
```

**** This file inputs node numbers for intermediate nodes***

```
$include dstrct_n.inc
```

**** This file inputs Israeli districts including intermediate nodes***

```
$include dstrctin.inc
```

**** The scalars and parameters (i.e. terms with fixed values—they***
**** are not changed in the optimization) used in the model are defined below.***

```
SCALARS

   MAXP        Cutoff price /100.0/
   MINP        Cutoff price /0.01/
   ISR2JERf    Fraction Israel supplies Jer fresh water (remainder by PNA)
   ISR2JERr    Fraction Israel supplies Jer recycled water (remainder by PNA)
   X           Yarmouk water coming in from Syria (0-500MCM)
   SUPMULT     Supply multiplier for drought and wet years
;

PARAMETERS

   POP(i1)       Population in district i1 for specified scenario
   ALPHA(i,d)    Exponent of inverse demand curve for each sector at district i
   QDL(i,d)      Water Demand by sector in node i in MCM for specified scenario
   B(i1,d)       Coefficient of inverse demand curve for each sector at district i in 1990
   P0(i1,d)      Price at current demands at district i in 1990 in $ per m3
   DEMMULT(i,d)  Demand Multiplier for district i and demand sector d, specified in VB
   QSMAX(i,s)    Upper Bound on Water Supply from Source s in node i in MCM
   CS(i,s)       Unit cost for supplying water from source s in node i in dollars per m3
```

```
INDD(i,s)        Indicates if supply source is in common pool
ARC(i1,j1)       Connections - 1 if arc is possible between i and j and 0 otherwise
CTR(i1,j1)       Unit transfer cost from node i to j - in cents per m3
DL(i1)           District leakage rates as a fraction if flow in district i
TL(i1,j1)        Inter-district transfer leakage rates as a fraction
ARCR(i1,j1)      Possible connections for recycled water in a parallel network
CTRY(i1,j1)      Unit cost to transfer recycled water from district i to j in dollars per m3
PRMAX(i1,d)      Maximum percent recycling at i from use d
CR(i1,d)         Unit cost to recycle water at node i in dollars per cubic meter
CE(i,d)          Environmental cost for sector d in district i in $ per cubic meter
SUBAMT(i,d)      Subsidy amount in dollars per m3 for agricultural water in i
SUBRATE(i,d)     Subsidy rate in percent for agricultural water in i
QLO(i,d)         Lower bound on quantity demanded
QMAX(i,d)        Upper bound on quantity demanded
;

* The following statements add files which include all of the
* definitions for the scalars and parameters declared above.

* This file gives the fraction Israel supplies Jerusalem fresh and
* recycled water, with the remainder by supplied by the PNA
$include "isrtojer.inc"

* Yarmouk boundary condition, as specified in Visual Basic by user (X)
$include "yarmouk.inc"

* DEMAND DEFINITIONS

* This file gives the population in district i1 for the
* specified scenario in Visual Basic (VB)
$include "pop.inc"

* This file includes the exponent of inverse demand curve for
* each sector in district i as specified in VB by user
$include "alpha2.inc"

* This file gives the price at current demands in district i in 1990 ($/m3)
$include "prices.inc"

* This file includes the water demand by sector in node i in MCM
* for specified scenario in VB
$include "dem.inc"

* This file includes a multiplier of demand function specified in VB (default is 1)
$include "demmult.inc"

* The demands are modified by any demand multipliers

  QDL(i,d) = QDL(i,d)*DEMMULT(i,d);
```

** The coefficients for the demand function is calculated from alpha*
** the price and quantity given for each district i and demand sector d*

```
B(i,d)$(QDL(i,d) gt 1e-3) = P0(i,d)/(QDL(i,d)**ALPHA(i,d));
B(i,d)$(QDL(i,d) le 1e-3) = 1e-5;
```

** SUPPLY DEFINITIONS*

** This file contains the upper bound on water supply in MCM from source s*
** in district i with the associated unit costs in $/CM*
$include "sup.inc"

** This file includes a multiplier of supplies specified in VB (default is 1)*
$include "supmult.inc"

** The supplies are modified by any supply multipliers*

```
QSMAX(i,s) = QSMAX(i,s)*SUPMULT;
```

** This file indicates which supplies are common pool resources (1 if yes,*
** 0 if no)*
$include "commonpl.inc"

** CONVEYANCE DEFINITIONS*

** This file gives the connections - 1 if arc is possible between i and j and 0 otherwise*
$include "arc.inc"

** This file gives the unit transfer cost from node i to j -in cents per m3*
$include "ctr.inc"

** The transport costs are converted into dollars per m3*

```
CTR(i1,j1) = CTR(i1,j1)*0.01;
```

** The transport costs for recycled links are assumed to be the same as for fresh*

```
CTRY(i1,j1) = CTR(i1,j1);
```

** This file provides the fraction of intra-district leakage*
** and inter-district transfer leakage rates as a fraction*
$include "leakage.inc"

** RECYCLING DEFINITIONS*

** This file gives the maximum percent recycling at i from use d*
$include "prmax.inc"

** This file gives the unit cost to recycle water at node i in dollars per m3*

```
$include "cr.inc"

* ENVIRONMENTAL COST DEFINITIONS

* This file gives the environmental cost for sector d in district i in $/m3
$include "ce.inc"


* The following declares the decision variables for the optimization problem

VARIABLES
   Z          Net Economic Benefit in Million Dollars
   QS(i1,s)   Quantity supplied by source s in district i in MCM
   QD(i1,d)   Quantity demanded by sector d in district i in MCM
   QTR(i1,j1) Quantity of water transferred from district i to district j in MCM
   QRY(i1,d)  Quantity of water recycled from use d at district i in MCM
   QFRY(i1,d) Quantity of recycled water supplied to use d at district i in MCM
   QTRY(i1,j1) Quantity of recycled water transferred from district i to district j in MCM
   PR(i1,d)   Percent recycled at district i from use d
;

POSITIVE VARIABLES QS,QD,QTR,PR,QRY,QFRY,QTRY;

* The following files modify the defaults defined above through
* user input in Visual Basic. The first set of files modify
* the system infrastructure. The second set of files modify
* the objective function or the constraints to represent different
* policy objectives.

* MODIFICATIONS TO INFRASTRUCTURE

* This file specifies which infrastructure option is selected:
* Current, Current+ or Plausible+
$include "arcc.inc"

* This file specifies which PLS Link option is chosen: 1,2,3,4 or 5
$include "option.inc"

* This file includes any desalination plants specified by the user
* in VB
$include "d.inc"

* This files modifies recycling plants on-line as chosen from VB menu
$include "recyplnt.inc"
$include "recyp.inc"

* These files specify modifications to links for fresh and recycled water
$include connect.inc
$include connecty.inc
$include "ctradd.inc"       {Additional cons for fresh wat added links}
```

National policies can be of one (but not more than one) of three types:

1. *The first of these is a percentage subsidy. In this case, the price paid by private consumers is equal to that "paid" by the government times (1 – SUBRATE/100). (SUBRATE is expressed in percentage points.) Equivalently, the price on the government's demand curve corresponding to any quantity is the private price ($P = BQ^{\alpha}$) divided by (1 – SUBRATE/100). This is the adjustment in the first term of the objective function. (Of course, if there is no percentage subsidy, SUBRATE = 0.)*

2. *The second kind of policy is a fixed amount per cubic meter. In that case, the social value of water per cubic meter exceeds the private value by the subsidy amount (and the government's demand curve is obtained by raising the private one uniformly by that amount). The second term in the objective function (the one involving SUBAMT) accounts for this shift.*

3. *The third kind of national policy is a fixed price policy (usually called "quotas" in the program. In this case, the government offers consumers a fixed price for water (in one or more steps), as in the following diagram:*



Figure 1

Here, it is as though the government had a vertical demand curve at Q (since it stands ready to supply consumers that quantity at the fixed price, P*. The area under (to the left of) that curve is fixed. (It is a rectangle whose top – not shown –is given by the maximum cost that that the government would be willing to incur to effect its policy.) Hence there is no need to include that area in the objective function. The only benefits that matter are still the benefits to the consumers – the area under their demand curve to the left of Q*, and these are already included in the objective function. Note that this remains true in the following diagram, where the demand curve does not cross the fixed-price-policy curve.*

```
$include "ctryadd.inc"      {Additional cons for recy wat added links}

* MODIFICATIONS FOR POLICY OBJECTIVES

* This file includes any subsidies in %
$include subrate.inc

* This file includes any subsidies in amount ($/m3)
$include subamt.inc

* These files incorporate information regarding fixed price policy,
* as specified in VB, and as calculated from quotas5.gms
$include quotas.inc
$include qindic.inc
$include reimburs.inc
$include hitindic.inc
$include p.inc

* The following statements define the equations that will be used
* in the optimization, both the objective function and the constraints.

EQUATIONS
OBJ            Net Econonomic Benefit in Million Dollars
CONT(i1)       Continuity equation at district i to balance supply and demand
CONTR(i1)      Continuity equation for recycling at district i to balance supply and demand
DEFNR(i1,d)    Definition of amount of water recycled
QLOBND(i,d)    Lower Bound on Qty demanded
QUPBND(i,d)    Upper Bound on Qty demanded
YARMOUK1       Yarmouk continuity constraint
QUOTAS1(i,d)   Quota constraint for quota districts
QUOTAS2(i,d)   Quota constraint for quota districts NUMBER 2

* This includes additional equations as specified in VB.

$include "aqeqn.inc"
```

The objective function begins with the sum over all districts of the integral under the demand curve (adjusted for national policies as described in a minute). The demand curve is represented by the equation: $P = BQ^{\alpha}$, and the integral is over Q. Note that the integral cannot be taken from zero to the quantity demanded, since, with an inelastic demand curve ($-1 < \alpha$), the area would be infinite. Hence we take the integral only from the quantity (QLOW) corresponding to the cutoff price (PMAX). Since the evaluation of the integral at QLOW is just a constant, the term corresponding to that is omitted from the objective function.

Where there are national policies, the private demand curve is adjusted to reflect the additional social value of water as revealed by those policies. A good way to think about this is to think of the government as purchasing the water and then passing it on to consumers. In such a case, we need to use the area under the government's demand curve.

```
(sum(s,QS(i1,s)) +
sum(j1$ARC(j1,i1),QTR(j1,i1)*(1-TL(j1,i1)))-
sum(j1$ARC(i1,j1),QTR(i1,j1))) *
(1-DL(i1));
```

```
CONTR(i1)..  sum(d,QFRY(i1,d))=E=sum(d,QRY(i1,d))+sum(j1$ARCR(j1,i1),QTRY(j1,i1)*(1-
                        TL(j1,i1)))-sum(j1$ARCR(i1,j1),QTRY(i1,j1));
```

The next equation states that the quantity recycled from a particular user class in a particular district equals the percentage recycled times the consumption of that user class.

```
DEFNR(i1,d).. QRY(i1,d) =E= PR(i1,d)*(QD(i1,d)+QFRY(i1,d));
```

The next two equations place upper and lower bounds on consumption.

```
QLOBND(i,d).. QD(i,d)+QFRY(i,d) =G= QLO(i,d);

QUPBND(i,d).. QD(i,d)+QFRY(i,d) =L= QMAX(i,d);
```

The next constraint limits total extractions from the Yarmouk (in two locations in Jordan District 1, one Israeli, and one Palestinian district) to the annual flow south of Syria. The shadow price here will be the shadow price of Yarmouk water. Following that, the other constraints on how much Yarmouk water can be taken at each location are given (again).

```
YARMOUK1..    QS("I3Y","S1")+QS("J1","S2")+QS("J1","S3")+QS("P10","S4")=L=X;
```

The next two lines involve the treatment of fixed-price policies. The first ensures that the amount consumed is not less than that which is demanded given from the government at fixed prices (See Figures 1 and 2, above). The second states that if the demand curve hits the fixed-price policy as in Figure 1, then the corresponding amount is what is consumed.

```
QUOTAS2(i,d).. (QD(i,d)+QFRY(i,d)) =G= QQ(i,d);

QUOTAS1(i,d)$(HITINDIC(i,d) ne 0)..   (QD(i,d)+QFRY(i,d)) =E= QQ(i,d);
```

This statement declares that all of the above scalars, sets, parameters, variables and

equations are part of the WAS model.

```
MODEL WAS /ALL/;
```

****BOUNDS -- Definitional Bounds***

```
QMAX(i,d) = 1E6;              { Maximum quantity in any supply step is 1E6}
QLO(i,d) = (MAXP/B(i,d))**(1/ALPHA(i,d)); {Lower bound on quantity demanded }
QS.UP(i,s) = QSMAX(i,s);        { Upper limit on Supply at each district for each source }
QS.FX(in,s) = 0;                         { No supply in intermediate districts    }
QD.FX(in,d) = 0;                          { No demand in intermediate districts    }
PR.UP(i1,d) = PRMAX(i1,d);             { Upper Bound for Wastewater Recycling    }
```



Figure 2

We next subtract from the objective function: the costs of extraction for all supply steps used; the cost of conveyance of fresh water for all fresh water conveyance links; the costs of recycling (above the level necessary for environmentally-safe disposal); the cost of conveyance for recycled water; and the environmental charge. (Note that desalination plants are treated as providing extra supply steps, so that their cost is included as though it were an extraction cost.)

```
OBJ..      Z =E=
        sum((i,d),(B(i,d)/((ALPHA(i,d)+1) *
        (1-SUBRATE(i,d)/100))*(QD(i,d)+QFRY(i,d))**(ALPHA(i,d)+1)
        + SUBAMT(i,d)*(QD(i,d)+QFRY(i,d))))
        - sum((i,s),(QS(i,s)*CS(i,s)))
        - sum((i1,j1)$ARC(i1,j1),(QTR(i1,j1)*CTR(i1,j1)))
        - sum((i1,d),(QRY(i1,d)*CR(i1,d)))
        - sum((i1,j1)$ARCR(i1,j1),(QTRY(i1,j1)*CTRY(i1,j1)))
        - sum((i,d),(QD(i,d)+QFRY(i,d))*CE(i,d));
```

We now deal with various constraints. The first of these is the constraint that the amount of fresh water consumed in any location must equal the amount extracted there plus the amount brought in from other locations minus the amount conveyed to other locations. The next constraint does the parallel thing for recycled water. Note that it is the shadow prices associated with these constraints that will be the shadow prices of the two types of water at the different locations. Note further that these constraints apply also to intermediate nodes where consumption and supply are constrained to be zero.

```
CONT(i1)..   sum(d,QD(i1,d)) =E=
```

```
PR.FX(in,d) = 0;                  { No recycling in intermediate districts }
QRY.FX(in,d) = 0;                 { No recycling in intermediate districts }
QFRY.FX(in,d) = 0;                { No recycling in intermediate districts }
QRY.FX(i1,"AGR") = 0;             { No recycling of agricultural water     }
QFRY.FX(i1,"URB") = 0;            { No recycled water for urban use        }
QFRY.FX(i1,"IND") = 0;            { No recycled water for industrial use   }
QD.FX(i,d)$(QDL(i,d) le 1e-3) = 1e-4;  {If there is no demand in a sector d, the minimum value is 1e-4 }
QD.L(i,d) = 1;          {The initial value for all demands is one }
QD.UP(i,d) = 1E6;   {The upper bound on quantity demanded is 1E6}
```

**** These should be removed....***

```
QS.UP("J1","S3") = X;
QS.UP("P10","S4") = X;
QS.UP("J1","S2")$(X LE QSMAX("J1","S2")) = X;
QS.UP("I3Y","S1")$(X GT 90) = 90;  {Shamir 4/21/95 40 to I3 and 50 local}
QS.UP("I3Y","S1")$(X LE 90) = X;
```

Note the order in which include files involving conveyance links are read in. The last of these to be read is the one that tells the program whether the model is to be countrified. If you choose a national model, you cannot then put in links connecting the country in question to the others.

```
$include country.inc
$include gazacon.inc
$include gazanort.inc
$include ramconn.inc
```

**** SOLVE!***

```
SOLVE WAS MAXIMIZING Z USING NLP;
```

**** The following definitions of parameters applies to post-processing***
**** calculations***

```
PARAMETERS

  PC(i,d)     Consumer Price in district i1 for use d
  PP(i,d)     Producer Price in district i1 for use d
  PE(i,d)     Producer Price for rec water if any in district i1 for use d
  PD          Average Value of water in dispute
  PY          Average Value of Yarmouk water
  NCS(i1)     National Consumer Surplus in district i1
  GC(i1)      Government Cost in district i1
  PS(i1)      Producer Surplus in district i1
  VALUE(i)    Value of water in each district
  REC_PRFT(i,d) Recyling Profit
  UAF_ENVC(i) Unaccounted-for environmental costs
  PEFF(i,d)   Price effectively paid by government
  REDLIGHT(i,d) Indicator for perverse results in fixed price policy
  PAPP(i1,d)  Appropriate price for district i1 for use d {this turns out to be the actual price paid if there
```

are no quotas or subsidies}

;

We now define various prices.

1. ***The first of these (PE) is the price that corresponds to the consumed quantity according to the consumers' demand curve. Where there is no national policy, this will be the price consumers actually pay.***
2. ***The second price (PP) is the shadow price of fresh water.***
3. ***So far as I can tell, the third (PI) is never used and should be deleted.***

The price, PC, will be the same as PE in the absence of a fixed-price policy. In its presence, it is the price that consumers actually pay.

```
PE(i,d) = B(i,d)*(QD.L(i,d)+QFRY.L(i,d))**ALPHA(i,d);
PP(i,d) = CONT.M(i);
PC(i,d) = B(i,d)*((QD.L(i,d)+QFRY.L(i,d))**ALPHA(i,d));
```

**** The following line sets the correct price when the quota steps don't hit***
**** the demand function***

```
PC(i,d)$((HITINDIC(i,d) eq 0) AND
     (QINDIC(i,d) eq 1) AND
     (QD.L(i,d)+QFRY.L(i,d) eq QQ(i,d))) = P(i,d,"5");
```

**** The following line sets the correct price when the quota steps cross***
**** the demand function***

```
PC(i,d)$((HITINDIC(i,d) eq 12) AND
     (QINDIC(i,d) eq 1) AND
     (QD.L(i,d)+QFRY.L(i,d) eq QQ(i,d))) = P(i,d,"1");
PC(i,d)$((HITINDIC(i,d) eq 13) AND
     (QINDIC(i,d) eq 1) AND
     (QD.L(i,d)+QFRY.L(i,d) eq QQ(i,d))) = P(i,d,"2");
PC(i,d)$((HITINDIC(i,d) eq 14) AND
     (QINDIC(i,d) eq 1) AND
     (QD.L(i,d)+QFRY.L(i,d) eq QQ(i,d))) = P(i,d,"3");
PC(i,d)$((HITINDIC(i,d) eq 15) AND
     (QINDIC(i,d) eq 1) AND
     (QD.L(i,d)+QFRY.L(i,d) eq QQ(i,d))) = P(i,d,"4");
```

We now calculate net benefits. These are divided into buyer (or consumer) surplus and producer surplus (or "profits"). To understand this, consider Figure 3. Here it is important to note that the step-like curve does not represent a fixed-price policy (as in Figures 1 and 2) but rather the marginal cost of supplying water. (The fact that it is drawn as a step function does not matter,) The area, A, is the amount of buyer surplus; it is the total benefit consumers receive minus the amount they pay to receive it. The rectangle, B, is the amount of producer

surplus; it is the total amount that producers receive for water minus the cost of supplying that water.



Figure 3

NOTE: The present version of the program calculates producer surplus ("profits") and allocates them to the district in which the relevant water is first extracted. One of the output pages (the "social welfare" page) shows this allocation. When running a national model, this makes no difference. When running a regional model, however, it matters if the user wishes to allocate benefits to countries. (It does not matter if all that is looked at is total benefits for the region.) Such an allocation requires allocating producer surplus by country of ownership rather than by district of extraction. In the present version of the program, that allocation has to be done off-line as a side calculation. In WAS 3.3, we intend to automate it.

The calculation of buyer surplus, then, begins in much the same way as the setting of the objective function, above, a fact that should surprise nobody. The first three lines give that calculation. They differ from the corresponding terms in the objective function only in that the evaluation of the integral at the lower limit of integration is subtracted. That lower limit is not zero, however, but the quantity (QLOW) that corresponds to the pre-set very high cut-off price. (Note that this fact means that one cannot attach meaning to the <u>level</u> of consumer surplus or the level of social welfare but only to <u>changes</u> in those magnitudes produced by different infrastructures or policies. Of course, this is all that matters.) The remaining terms in the equation give the additional surplus given to consumers if they pay a price (PC) less than the price (PE) that corresponds to the quantity consumed according to their demand curve. This can happen in the case of a fixed-price policy as illustrated in Figure 4, where the extra surplus is the area of the rectangle, C.



Figure 4

```
NCS(i) = sum(d,((B(i,d)/((ALPHA(i,d)+1)*(1-SUBRATE([illegible]
    ((QD.L(i,d)+QFRY.L(i,d))**(ALPHA(i,d)+1)-QLO(i,d)**(ALPHA(i,d)+1))
    +SUBAMT(i,d)*(QD.L(i,d)+QFRY.L(i,d)-QLO(i,d))
    +QINDIC(i,d)*(PE(i,d)-PC(i,d))*(QD.L(i,d)+QFRY.L(i,d))
    -PE(i,d)*(QD.L(i,d)+QFRY.L(i,d)) ));
```

We now calculate the effective price paid by shadow price for each type of user. For those who do not use recycled water, this is the shadow price of fresh water. For those who do, it is the shadow price of recycled water. Note that, in the current version of the program, agricultural users are assumed to use recycled and fresh water with substitution occurring on a one-to-one basis. This means that recycled water can never have a higher shadow price than fresh water and will have the same shadow price if both are being used by agriculture in the same district.

It has been suggested that we allow for the possibility that recycled water is preferred to fresh in agriculture for its nutrient value. In WAS 3.3, this is most conveniently done by assuming that one cubic meter of recycled water does the same job in agriculture as b cubic meters of fresh water, where b is less than unity. If that is done, the second statement below will have to be altered.

```
PEFF(i,d)$(QFRY.L(i,d) eq 0) = CONT.M(i);
PEFF(i,d)$(QFRY.L(i,d) ne 0) = CONTR.M(i);
```

The next equation defines recycling profit per cubic meter of fresh water available for recycling. It is the shadow price of recycled water per cubic meter less the corresponding cost times the percentage of the user's fresh water that is recycled. This profit will be credited to the fresh water user as an offset to the environmental charge.

```
REC_PRFT(i,d) = (CONTR.M(i) - CR(i,d))*PR.L(i,d);
```

The next equation defines otherwise unaccounted-for environmental costs. These occur when there are fixed-price policies. In such a case, consumers pay the fixed prices (on the quantities to which they apply, totaling QQ). But then the environmental charge (adjusted for recycling profits) on those quantities is paid by nobody. (This is not true for the other two types of

policies, in which the government can be thought of as paying the price that consumers would otherwise pay – including the environmental charge – and then reselling to consumers at the subsidized price.) Since the environmental charge represents a real cost to the system, it must be subtracted from social welfare even if it does not show up explicitly in the cash accounting.

```
UAF_ENVC(i) = sum(d, Qindic(i,d)*(CE(i,d)-REC_PRFT(i,d))*QQ(i,d));
```

We now calculate government financial costs when there is a national policy. This amounts to taking the difference between the amount that consumers would pay without such a policy and subtracting the amount that they actually do pay. In the case of the two subsidy policies, the price that consumers would pay in the absence of the policy is the shadow price of the water in question plus the environmental charge adjusted for recycling profits. PC is the price they actually do pay. In the case of a fixed price policy, the government is taken as paying the shadow price and receiving reimbursement from the consumers. (This is the case in which environmental costs are unaccounted for. In the other cases, the government effectively lets the consumer pay the price that includes the environmental charge and then gives the consumer a rebate.)

```
GC(i) = sum(d, (1-Qindic(i,d))*(((PP(i,d)+CE(i,d)-REC_PRFT(i,d)-PC(i,d))*QD.L(i,d))
            + (CONTR.M(i)-PC(i,d))*QFRY.L(i,d))
            + Qindic(i,d)*(QQ(i,d)*PEFF(i,d)-Reimburs(i,d)));
```

The following lines produce a warning if an incomplete price policy has created an ambiguity because "market" price has turned out to be below the fixed prices.

```
PAPP(i,d)=PP(i,d)+CE(i,d)-REC_PRFT(i,d);
PAPP(i,"AGR")=CONTR.M(i);
$include "qqtot.inc"
REDLIGHT(i,d)=0;
REDLIGHT(i,d)$((HITINDIC(i,d) ne 0) AND
        (QQTOT(i,d) lt 999) AND
        (QQTOT(i,d) lt (PAPP(i,d)/B(i,d))**(1/ALPHA(i,d)) ) AND
        (PC(i,d) gt PAPP(i,d)))=1;
$include "red.inc"
```

The next line calculates producer surplus on fresh water by district. (The shadow price is chosen for "AGR" because GAMS can't take a general index here and shadow price is independent of user type, anyway.

```
PS(i) = sum(s,(QS.L(i,s)*(PP(i,"AGR")-CS(i,s))));
```

The next line calculates buyer expenditure on water. (There is a slight error in the present version in that PC rather than PE should be used. This will affect only some fixed-price cases. In any event, VALUE is simply reported in the output and not used to find the solution.)

```
VALUE(i) = sum(d,((QD.L(i,d)+QFRY.L(i,d))*(PE(i,d))));
```

The next line gives the shadow price of Yarmouk water in situ.

```
PY = YARMOUK1.M;
```

The remaining lines are for generating the output for reading into Visual Basic.

```
$include "pddef.inc"
$include "end.inc"
$include "end2.inc"
```